BenderOrszag Ex2p283 04-11-16

N. T. Gladd

Initialization: Be sure the files *NTGStylesheet2.nb* and *NTGUtilityFunctions.m* is are in the same directory as that from which this notebook was loaded. Then execute the cell immediately below by mousing left on the cell bar to the right of that cell and then typing “shift” + “enter”. Respond “Yes” in response to the query to evaluate initialization cells.

In[56]:=
```
SetDirectory[NotebookDirectory[]];
(* set directory where source files are located *)
SetOptions[EvaluationNotebook[], (* load the StyleSheet *)
 StyleDefinitions → Get["NTGStylesheet2.nb"]];
Get["NTGUtilityFunctions.m"]; (* Load utilities package *)
```

Original notebook — *BenderOrszag Ex2p283 09-14-15*

Purpose

This example of the steepest descent method involves the integral

$$\mathcal{I}(k) = \int_0^1 dx\, e^{i k x^2} \tag{1}$$

Mathematica finds the closed form

In[120]:=
```
w0[1] = Integrate[Exp[I k x^2], {x, 0, 1}]
```

Out[120]=

$$-\frac{(-1)^{1/4}\sqrt{\pi}\,\mathrm{Erf}\left[(-1)^{3/4}\sqrt{k}\right]}{2\sqrt{k}}$$

for which the large k expansion is

In[59]:=
```
w0[2] = Series[w0[1], {k, ∞, 3}]
```

Out[59]=

$$\left(\frac{1}{2}(-1)^{1/4}\sqrt{\pi}\sqrt{\frac{1}{k}} + O\left[\frac{1}{k}\right]^{7/2}\right) + e^{i k}\left(-\frac{i}{2k} - \frac{1}{4k^2} + \frac{3i}{8k^3} + O\left[\frac{1}{k}\right]^4\right)$$

This integral may also be evaluated by integration by parts. However, the straightforward approach

$$\mathcal{I}(k) = \int_0^1 dx\, e^{i k x^2} = \int_0^\infty dx \left(\frac{1}{2 i k x}\right)\left(\frac{d\, e^{i k x^2}}{d x}\right) \tag{2}$$

doesn’t work because of a singularity occurs at the lower limit.

This problem can be avoided by writing

$$\mathcal{I}(k) = \int_0^1 dx\, e^{i k x^2} = \int_0^\infty dx\, e^{i k x^2} - \int_1^\infty dx\, e^{i k x^2} \tag{3}$$

and noticing that the first integral can be evaluated

In[60]:=
```
w0[1] = Simplify[Integrate[Exp[I k x^2], {x, 0, ∞}], {Im[k] > 0}]
```

Out[60]=

$$\frac{\sqrt{\pi}}{2\sqrt{-i k}}$$

So

$$\mathcal{I}(k) = \frac{\sqrt{\pi}}{2\sqrt{-i k}} - \int_1^\infty dx\, e^{i k x^2} \tag{4}$$

I Evaluation of $\mathcal{I}$(k) by integration by parts

For the purpose of integrating by parts I note that

$$\mathcal{I}(k) = \frac{\sqrt{\pi}}{2\sqrt{-i k}} - \int_1^\infty dx\, e^{i k x^2} = \frac{\sqrt{\pi}}{2\sqrt{-i k}} - \int_1^\infty dx \left(\frac{1}{2 i k x}\right)\left(\frac{d\, e^{i k x^2}}{d x}\right) \tag{5}$$

and write a function that performs the integration by parts

In[61]:=
```
Clear[IntByParts];
IntByParts[f_, Ustart_, dV_, {x_, l_, u_}] :=
 Module[{U = f Ustart, dU, V},
  dU = D[U, x];
  V = Integrate[dV, x];
  (*Print["{f, U, dU, V, dV } = ",{f, U,dU, V, dV}];*)
  Limit[(U V /. x → u), Assumptions → {x ∈ Reals, x > 0, Im[k] > 0}] -
   (U V /. x → l) - Int[V dU, {x, l, u}]]
```

For example,

In[63]:=
```
w1[1] = With[{f = 1},
  IntByParts[f, Exp[I k x^2] / D[Exp[I k x^2], x], D[Exp[I k x^2], x], {x, 1, ∞}] ]
```

Out[63]=

$$\frac{i\, e^{i k}}{2 k} - \text{Int}\left[\frac{i\, e^{i k x^2}}{2 k x^2}, \{x, 1, \infty\}\right]$$

where the next integration only involves a new value of f.

Successive integrations by parts can be accomplished with the recursion

In[64]:=
```
Clear[AsymptoticSeries];
AsymptoticSeries[1] =
  IntByParts[1, Exp[I k x^2] / D[Exp[I k x^2], x], D[Exp[I k x^2], x], {x, 1, ∞}];
(* note use of memoization so that lower terms are remembered *)
AsymptoticSeries[n_ /; n > 1] := AsymptoticSeries[n] =
   With[{f = AsymptoticSeries[n - 1][[-1]] /. Int[a_ Exp[I k x^2], b_] → a},
    AsymptoticSeries[n - 1][[1 ;; -2]] -
     IntByParts[f, Exp[I k x^2] / D[Exp[I k x^2], x], D[Exp[I k x^2], x], {x, 1, ∞}]];
```

In[67]:=
```
AsymptoticSeries[#] & /@ Range[1, 5] // ColumnForm
```

Out[67]=

$$\frac{i e^{i k}}{2 k} - \text{Int}\left[\frac{i e^{i k x^2}}{2 k x^2}, \{x, 1, \infty\}\right]$$
$$-\frac{e^{i k}}{4 k^2} + \frac{i e^{i k}}{2 k} + \text{Int}\left[\frac{3 e^{i k x^2}}{4 k^2 x^4}, \{x, 1, \infty\}\right]$$
$$-\frac{3 i e^{i k}}{8 k^3} - \frac{e^{i k}}{4 k^2} + \frac{i e^{i k}}{2 k} + \text{Int}\left[\frac{15 i e^{i k x^2}}{8 k^3 x^6}, \{x, 1, \infty\}\right]$$
$$\frac{15 e^{i k}}{16 k^4} - \frac{3 i e^{i k}}{8 k^3} - \frac{e^{i k}}{4 k^2} + \frac{i e^{i k}}{2 k} + \text{Int}\left[-\frac{105 e^{i k x^2}}{16 k^4 x^8}, \{x, 1, \infty\}\right]$$
$$\frac{105 i e^{i k}}{32 k^5} + \frac{15 e^{i k}}{16 k^4} - \frac{3 i e^{i k}}{8 k^3} - \frac{e^{i k}}{4 k^2} + \frac{i e^{i k}}{2 k} + \text{Int}\left[-\frac{945 i e^{i k x^2}}{32 k^5 x^{10}}, \{x, 1, \infty\}\right]$$

For future convenience I define

In[68]:=
```
Clear[ℱexact, ℱasymptotic];
ℱexact[k_] := -((-1)^(1/4) Sqrt[π] Erf[(-1)^(3/4) Sqrt[k]]) / (2 Sqrt[k]);
ℱasymptotic[n_] := Sqrt[π] / (2 Sqrt[-i k]) - AsymptoticSeries[n][[1 ;; n]]
```

$$\mathcal{F}\text{exact}[k\_] := -\frac{(-1)^{1/4}\sqrt{\pi}\,\text{Erf}\left[(-1)^{3/4}\sqrt{k}\right]}{2\sqrt{k}};$$
$$\mathcal{F}\text{asymptotic}[n\_] := \frac{\sqrt{\pi}}{2\sqrt{-i k}} - \text{AsymptoticSeries}[n][[1 ;; n]]$$

For example,

In[71]:=
```
{ℱasymptotic[1], ℱasymptotic[2], ℱasymptotic[3]}
```

Out[71]=

$$\left\{-\frac{i e^{i k}}{2 k} + \frac{\sqrt{\pi}}{2\sqrt{-i k}}, \frac{e^{i k}}{4 k^2} - \frac{i e^{i k}}{2 k} + \frac{\sqrt{\pi}}{2\sqrt{-i k}}, \frac{3 i e^{i k}}{8 k^3} + \frac{e^{i k}}{4 k^2} - \frac{i e^{i k}}{2 k} + \frac{\sqrt{\pi}}{2\sqrt{-i k}}\right\}$$

I compare the exact value with the leading order asymptotic form

In[73]:=
```
Plot[{Re@ℱexact[k], Im@ℱexact[k], Re@ℱasymptotic[1], Im@ℱasymptotic[1]},
 {k, 1, 5}, AxesLabel → {Stl["k"], Stl["ℐ(k)"]}, PlotLabel →
  Stl["Exact vs leading asymptotic term\nExact (black), asymptotic form (blue)"],
 PlotStyle → {Black, Directive[Black, Dashed], Blue,
   Directive[Blue, Dashed], Green, Directive[Green, Dashed]}]
```

Out[73]=



2 Evaluation by steepest descent

Following the development in Bender and Orszag, I also use the steepest descent method to value ℐ(k).

I begin by visualizing the function over which the integral is performed. (see below for code that creates the following graphic)



It is instructive to view |F(x, k)|.

In[75]:=
```
Module[{κ = 2, Y = 2, X = 2, image = 300, gSurface, lab, F},
 F[z_, k_] := Exp[I k z^2];

 lab =
  Stl@StringForm["k = ``  F(z, k) = ``", κ, HoldForm@TraditionalForm[Exp[I k z^2]]];
 gSurface = Plot3D[Abs[F[x + I y, κ]], {x, -X, X}, {y, -Y, Y},
   ImageSize → image, MeshFunctions -> {#3 &}, Mesh → 20, Boxed → False,
   AxesLabel → {Stl["x"], Stl["y"], Stl["|F(z)|"]}, PlotLabel → lab]]
```

Out[75]=



A "standard" representation of the integrand, for the purpose of valuing the quantities needed for steepest descent, is

$$\exp(k\,\rho(z)) = \exp(k\,(\rho(x + i\,y)\}) = \exp(k\,(\phi(x,\ y) + i\,\psi(x,\ y)) \qquad (6)$$

The starting and ending points of the integration path are of particular interest. For the steepest descent path originating from z = 0 + I 0.

In[87]:=
```
A00 = ConstructSteepestDescentAssociation[Exp[I k z^2], 0];
LGrid[({#[[1]], #[[2]]}) & /@ Normal[A00], "Steepest Descent Information at z = 0"]
```

Out[88]=

Steepest Descent Information at z = 0

F	$e^{i k z^2}$
ρ	$i z^2$
zChosen	0
ϕ	$-2 x y$
ψ	$x^2 - y^2$
constantPhaseEqn	$0 == x^2 - y^2$
constantPhaseCurves	$\{\{\{y \to -x\}, \phi \to 2 x^2\}, \{\{y \to x\}, \phi \to -2 x^2\}\}$

For the steepest descent path passing through z = 1 + i 0

In[89]:=
```
A10 = ConstructSteepestDescentAssociation[Exp[I k z^2], 1];
LGrid[({#[[1]], #[[2]]}) & /@ Normal[A10], "Steepest Descent Information at z = 1"]
```

Out[90]=

Steepest Descent Information at z = 1

F	$e^{i k z^2}$
ρ	$i z^2$
zChosen	1
ϕ	$-2 x y$
ψ	$x^2 - y^2$
constantPhaseEqn	$1 == x^2 - y^2$
constantPhaseCurves	$\{\{\{y \to -\sqrt{-1+x^2}\}, \phi \to 2x\sqrt{-1+x^2}\}, \{\{y \to \sqrt{-1+x^2}\}, \phi \to -2x\sqrt{-1+x^2}\}\}$

ConstructSteepestDescentAssociation is an approach to systematically calculating steepest descent curves for integrands with the standard form. This information, once calculated in stored in a Mathematica data structure called an Association.

In[91]:=
```
Clear[ConstructSteepestDescentAssociation];
ConstructSteepestDescentAssociation[F_, zChosen_] :=
 Module[{ρ, zRoot, ϕ, ψ, ρAtzChosen,
   constantPhaseEqn, solns, constantPhaseCurves, names, values},
  ρ = F /. Exp[a_] → a / k;
  ϕ = ComplexExpand[Re[ρ /. z → x + I y] ];
  ψ = ComplexExpand[Im[ρ /. z → x + I y] ];
  ρAtzChosen = ρ /. z → zChosen;
  constantPhaseEqn = Im[ρAtzChosen]  == ψ;
  solns = Solve[constantPhaseEqn, y];
  constantPhaseCurves = ({#, "ϕ" → (ϕ /. #)}) & /@ solns;
  names =
   {"F", "ρ", "zChosen", "ϕ", "ψ", "constantPhaseEqn", "constantPhaseCurves"};
  values = {F, ρ, zChosen, ϕ, ψ, constantPhaseEqn, constantPhaseCurves};
  AssociationThread[names, values]]
```

I visualize the two steepest descent paths



I show a possible deformation of the integration path that follows the steepest descent paths.



3 C_2 contribution for steepest descent.

First, I note that the contribution along C_2, which runs from Y + i Y to Y + i $\sqrt{1+Y^2}$ is zero as Y → ∞,

Along C_2 y = x or, in parametric form, z = t + i Y, dz = dt

In[95]:=
```
w3[1] = {z == (t + I Y), dz == dt}
```
Out[95]=
```
{z == t + i Y, dz == dt}
```

In[96]:=
```
w3[2] = Exp[I k z^2] dz /. (w3[1][[1]] // ER) /. (w3[1][[2]] // ER) /. dt → 1
```
Out[96]=

$$e^{i k (t+i Y)^2}$$

This can be valued

In[97]:=
```
w3[3] = Integrate[w3[2], {t, Y, Sqrt[1 + Y^2]}]
```
Out[97]=

$$-\frac{1}{2\sqrt{k}}(-1)^{1/4}\sqrt{\pi}\left(\text{Erf}\left[\sqrt{2}\sqrt{k}\,Y\right]+i\,\text{Erfi}\left[(-1)^{1/4}\sqrt{k}\left(i\,Y+\sqrt{1+Y^2}\right)\right]\right)$$

I expand about Y = ∞

In[98]:=
```
w3[4] = Normal@Series[w3[3], {Y, ∞, 1}] // Simplify
```

Out[98]=
$$\frac{\left(\frac{1}{4}+\frac{i}{4}\right)\left(e^{-2 k Y^2}-e^{k\left(i-2 Y \sqrt{1+Y^2}\right)}\right)}{k Y}$$

Take the limit as Y → ∞ and the C_2 contribution tends to 0.

In[99]:=
```
w3[5] = Limit[w3[4], Y → ∞, Assumptions → {k ∈ Reals, k > 0}]
```

Out[99]=
```
0
```

4 C_1 contribution for steepest descent.

Along C_1 y = x, or, in parametric form, z = (1 + i) t

In[100]:=
```
w4[1] = {z == (1 + I) t, dz == (1 + I) dt}
```

Out[100]=
$$\{z == (1+i)\, t,\ dz == (1+i)\, dt\}$$

In[101]:=
```
w4[2] = Exp[I k z^2] dz /. (w4[1][[1]] // ER) /. (w4[1][[2]] // ER) /. dt → 1
```

Out[101]=
$$(1+i)\, e^{-2 k t^2}$$

In[102]:=
```
w4["final"] = Integrate[w4[2] , {t, 0, ∞}, Assumptions → {Re[k] > 0}] /. dt → 1
```

Out[102]=
$$\frac{\left(\frac{1}{2}+\frac{i}{2}\right)\sqrt{\frac{\pi}{2}}}{\sqrt{k}}$$

5 C_3 contribution for steepest descent.

Along C_3 $y = \sqrt{x^2 - 1}$ or $z = t + i\sqrt{t^2 - 1}$

In[103]:=
```
w5[1] = z == t + I Sqrt[t^2 + 1]
```

Out[103]=
$$z == t + i\sqrt{1+t^2}$$

In[104]:=
```
w5[2] = dz == D[w5[1][[2]], t] dt
```

Out[104]=
$$dz == dt \left(1 + \frac{i\, t}{\sqrt{1+t^2}}\right)$$

In[105]:=
```
w5[3] = Exp[I k z^2] dz /. (w5[1] // ER) /. (w5[2] // ER) /. dt → 1 // Expand
```

Out[105]=
$$e^{i k \left(t+i \sqrt{1+t^2}\right)^2} + \frac{i\, e^{i k \left(t+i \sqrt{1+t^2}\right)^2} t}{\sqrt{1+t^2}}$$

Mathematica does not recognize either of these integrals.

In[106]:=
```
{Integrate[w5[3][[1]], {t, 0, ∞}], Integrate[w5[3][[2]], {t, 0, ∞}]}
```

Out[106]=
$$\left\{\int_0^\infty e^{i k \left(t+i \sqrt{1+t^2}\right)^2} dt, \int_0^\infty \frac{i\, e^{i k \left(t+i \sqrt{1+t^2}\right)^2} t}{\sqrt{1+t^2}} dt\right\}$$

BO cast these integrals into a form where Watson’s lemma can be evoked. As integrand of the form exp[-k u] is desired, which
suggests the choice of variables

In[107]:=
```
w5[4] = I k z^2 == k (I - u)
```

Out[107]=
$$i\, k\, z^2 == k\, (i - u)$$

or

In[108]:=
```
w5[5] = Map[(#/(I k)) &, w5[4], {1}] // Expand
```

Out[108]=
$$z^2 == 1 + i\, u$$

or

In[109]:=
```
w5[6] = Solve[w5[5], z][[2, 1]]
```

Out[109]=
$$z \to \sqrt{1 + i\, u}$$

In[110]:=
```
w5[7] = dz → D[w5[6][[2]], u] du
```

Out[110]=
$$dz \to \frac{i\, du}{2\sqrt{1 + i\, u}}$$

The integrand is

In[111]:=
```
w5[8] = dz Exp[I k z^2] /. w5[6] /. w5[7] /. du → 1 // ExpandAll
```

Out[111]=
$$\frac{i\, e^{i k - k u}}{2 \sqrt{1 + i u}}$$

Mathematica can evaluate this integral

In[112]:=
```
w5[9] = Integrate[w5[8], {u, 0, ∞}, Assumptions → {Re[k] > 0}]
```

Out[112]=
$$-\frac{(-1)^{1/4} \sqrt{\pi} \left(-2 + \text{Erfc}\left[(-1)^{3/4} \sqrt{k}\right]\right)}{2 \sqrt{k}}$$

However,to follow the Bender Orszag evocation of Watson's Lemma, a power series expansion of the radical is performed

In[113]:=
```
w5[10] = Normal@Series[1/Sqrt[1 + i u], {u, 0, 5}]
```

Out[113]=
$$1 - \frac{i u}{2} - \frac{3 u^2}{8} + \frac{5 i u^3}{16} + \frac{35 u^4}{128} - \frac{63 i u^5}{256}$$

In[114]:=
```
w5[11] = w5[8] /. 1/Sqrt[1 + i u] → w5[10] // Expand
```

Out[114]=
$$\frac{1}{2} i e^{i k - k u} + \frac{1}{4} e^{i k - k u} u - \frac{3}{16} i e^{i k - k u} u^2 - \frac{5}{32} e^{i k - k u} u^3 + \frac{35}{256} i e^{i k - k u} u^4 + \frac{63}{512} e^{i k - k u} u^5$$

Then

In[115]:=
```
w5["final"] = Integrate[#, {u, ∞, 0}, Assumptions → Re[k] > 0] & /@ w5[11]
```

Out[115]=
$$-\frac{945 e^{i k}}{64 k^6} - \frac{105 i e^{i k}}{32 k^5} + \frac{15 e^{i k}}{16 k^4} + \frac{3 i e^{i k}}{8 k^3} - \frac{e^{i k}}{4 k^2} - \frac{i e^{i k}}{2 k}$$

6 Comparison with exact form

The contributions from contours C_1 and C_2 are

In[116]:=
```
w6[1] = w4["final"] + w5["final"]
```

Out[116]=
$$-\frac{945 e^{i k}}{64 k^6} - \frac{105 i e^{i k}}{32 k^5} + \frac{15 e^{i k}}{16 k^4} + \frac{3 i e^{i k}}{8 k^3} - \frac{e^{i k}}{4 k^2} - \frac{i e^{i k}}{2 k} + \frac{\left(\frac{1}{2} + \frac{i}{2}\right) \sqrt{\frac{\pi}{2}}}{\sqrt{k}}$$

Recall from Section 1 that the original integral can actually be evaluated in this case

In[121]:=
```
w6[2] = w0[1]
```

Out[121]=
$$-\frac{(-1)^{1/4}\sqrt{\pi}\,\mathrm{Erf}\left[(-1)^{3/4}\sqrt{k}\right]}{2\sqrt{k}}$$

To compare with the steepest descent result I expand this expression for large k.

In[122]:=
```
w6[3] = Normal@Series[w6[2], {k, ∞, 6}]
```

Out[122]=
$$e^{i k}\left(-\frac{945}{64 k^6}-\frac{105 i}{32 k^5}+\frac{15}{16 k^4}+\frac{3 i}{8 k^3}-\frac{1}{4 k^2}-\frac{i}{2 k}\right)+\frac{1}{2}(-1)^{1/4}\sqrt{\frac{1}{k}}\sqrt{\pi}$$

In[123]:=
```
w6[4] = w6[3] - w6[1] // Simplify // PowerExpand
```

Out[123]=
```
0
```

and there is agreement. As expected, integration by parts and steepest descent generate the same asymptotic series.

Graphics

In[74]:=
```
Module[{κ = 2, imageSize = 400, lab, C, g, F, DirectiveList, CArrow},
 F[z_, k_] := Exp[I k z^2];
 DirectiveList[plotLabel_, {xLabel_, yLabel_, zLabel_}] :=
  Sequence[PlotLabel → plotLabel, AxesLabel → {xLabel, yLabel, zLabel},
   ImageSize → imageSize, MeshFunctions -> {#3 &}, Mesh → 5];
 CArrow[{p1_, p2_}, {lab_, offset_}] :=
  {PointSize[0.015], Point[p1], Point[p2], Arrow[{p1, (p1 + p2) / 2}],
   Line[{(p1 + p2) / 2, p2}], Style[Text[lab, offset + p1 + (p2 - p1) / 2], Small]};

 lab =
  Stl@StringForm["k = ``   F(z, k) = ``", κ, HoldForm@TraditionalForm[Exp[I k z^2]]];
 C = {BLACK, CArrow[{{0, 0}, {1, 0}}, {Stl["C"], {0, 0.25}}]};
 g[1] = Plot3D[Re[F[x + I y, κ]], {x, -2, 2},
   {y, -2, 2}, Evaluate[DirectiveList[lab, {"x", "y", "Re[F]"}]]];
 g[2] = ContourPlot[Re[F[x + I y, κ]], {x, -2, 2}, {y, -2, 2},
   AxesLabel → {"x", "y", ""}, ImageSize → 300, Epilog → C];
 g[3] = Plot3D[Im[F[x + I y, κ]], {x, -2, 2}, {y, -2, 2},
   Evaluate[DirectiveList["", {"x", "y", "Im[F]"}]]];
 g[4] = ContourPlot[Im[F[x + I y, κ]], {x, -2, 2}, {y, -2, 2},
   AxesLabel → {"x", "y", ""}, ImageSize → 300, Epilog → C];
 Grid[{{g[1], g[2]}, {g[3], g[4]}}]]
```

Out[74]=



In[93]:=

```
Module[{k = 2, xMin = -1, xMax = 3, yMin = -1, yMax = 3, zMin = 0,
  zMax = 2, imageSize = 300, lab, g, point0, sdpath0, point1, sdpath1,
  integrationPath, gCurve, curve, surface, F, PointOnPath, Path3D},
 F[z_, k_] := Exp[I k z^2];
 PointOnPath[x_, y_, k_] :=
  {x, y, Abs[F[x + I y, k]] + 0.01};
 Path3D[points_, text_] :=
  Module[{n = Length@points, nStart, nFin, nMid, arrow, txt},
   {nStart, nFin} = {Round[0.2 n], Round[0.25 n]};
   nMid = Round[0.5 n];
   arrow = Arrow[{points〚nStart〛, points〚nFin〛}];
   txt = Text[text, points〚nMid〛];
   {Line[points], arrow, txt}] ;

 (* {{y→-x},"ϕ"→2 x^2} *)
 point0 = Point[PointOnPath[0, 0, k]];
 sdpath0 =
  Path3D[Table[PointOnPath[x, x, k], {x, 0, 3, 0.01}], Style["", 16, Bold, Black]];
 (* {y→Sqrt[-1+x^2]},"ϕ"→2 x Sqrt[-1+x^2]} *)
 point1 = Point[PointOnPath[1, 0, k]];
 sdpath1 = Path3D[
   Table[PointOnPath[x, Sqrt[x^2 - 1], k], {x, 1, 3, 0.01}], Style["", 16, Bold, Black]];

 integrationPath = Path3D[
   Table[{x, 0, Abs[ Exp[I k x^2]]} , {x, 0, 1, 0.01}], Style["", 16, Bold, Black]];

 lab =
  Stl["Steepest descent paths running through\nz = 0 (Green) and z = 1 (Blue)"];
 gCurve = Graphics3D[{{Green, PointSize[0.03], point0, sdpath0},
    {Blue, PointSize[0.03], point1, sdpath1},
    {integrationPath}}, PlotRange → {{xMin, xMax}, {yMin, yMax}, {zMin, zMax}},
   Axes → Automatic, AxesLabel → {Stl["x"], Stl["y"], Stl["|F|"]},
   ImageSize → imageSize, Boxed → False, PlotLabel → lab];
 surface = Plot3D[Abs[F[x + I y, k]], {x, xMin, xMax}, {y, yMin, yMax},
   ImageSize → imageSize, MeshFunctions -> {#3 &}, Mesh → 10, Boxed → False,
   PlotRange → {{xMin, xMax}, {yMin, yMax}, {zMin, zMax}}];
 Show[{gCurve, surface}]]
```

Out[93]=


Steepest descent paths running through
z = 0 (Green) and z = 1 (Blue)
|F|
2.0
1.5
1.0
0.5
0.0
x
−1
0
1
2
3
y
−1
0
1
2
3

In[94]:=

```
Module[{Y = 1.5, P1, P2, C, C1, C2, C3, lab},
 {P1, P2} = {{Black, PointSize[0.02], Point[{0, 0}]},
   {Black, PointSize[0.02], Point[{1, 0}]}};
 C = {Directive[Black, Thick], Arrowheads[{0.0, 0.05, 0.05, 0.0}],
   Arrow[{{0, 0}, {1, 0}}], {Black, Stl@Text["C", {0.5, 0}]} };
 C1 = {Directive[Blue, Thick], Arrowheads[{0.0, 0.05, 0.05, 0.0}],
   Arrow[{{0, 0}, {Y, Y}}], {Black, Stl@Text["C1", {1, 1}]} };
 C2 = {Directive[Blue, Thick], Arrowheads[{0.0, 0.05, 0.05, 0.0}],
   Arrow[{{Y, Y}, {Sqrt[1 + Y^2], Y}}], {Black, Stl@Text["C2", {(Y + Sqrt[1 + Y^2])/2, Y}]} };
 C3 = Line@Table[{x, Sqrt[x^2 - 1]}, {x, 1, Sqrt[1 + Y^2], 0.01}];
 C3 = {Directive[Blue, Thick], Arrowheads[{0.0, 0.05, 0.05, 0.0}],
   Arrow[Table[{x, Sqrt[x^2 - 1]}, {x, Sqrt[1 + Y^2], 1, - 0.01}]],
   {Black, Stl@Text["C3", {1.4, Sqrt[1.4^2 - 1]}]} };
 lab = Stl@StringForm["Possible contour following steepest descent paths"];
 Graphics[{P1, P2, C, C1, C2, C3}, Axes → Automatic,
  AspectRatio → 1, AxesLabel → {Stl["x"], Stl["y"]},
  PlotRange → {{-0.25, 2}, {-0.25, 2}}, PlotLabel → lab]]
```

Out[94]=